Stata tip 39: In a list or out? In a range or out?
Nicholas J. Cox
Department of Geography
Durham University
Durham City, UK
n.j.cox@durham.ac.uk

Two simple but useful functions, `inlist()` and `inrange()`, were added in Stata 7, but users somehow still often overlook them. The manual entry `functions` gives formal statements on definitions and limits. The aim here is to emphasize with examples how natural and helpful these functions can be.

The question answered by `inlist()` is whether a specified argument belongs to a specified list. That answered by `inrange()` is whether a specified argument falls in a specified range. We can ask the converse question, of not belonging to or falling outside a list or range, by simply negating the function. Thus `!inlist()` and `!inrange()` can be read as “not in list” and “not in range”.

These functions can reduce your typing, reduce the risk of small errors, and make your Stata code easier to read and maintain. Thus with the `auto` data in memory, consider the choice for the integer-valued variable `rep78` between older ways of getting a simple listing,

```
. list make rep78 if rep78 == 3 | rep78 == 4 | rep78 == 5
. list make rep78 if rep78 >= 3 & rep78 <= 5
. list make rep78 if rep78 > 2 & rep78 < 6
```

and newer ways of getting the same listing,

```
. list make rep78 if inlist(rep78, 3, 4, 5)
. list make rep78 if inrange(rep78, 3, 5)
```

The examples here are typical of a good way to use `inlist()` or `inrange()`: move directly from feeding arguments to each function to using the results of the calculation. If you wanted to keep the results, you could put them into a variable (or a macro). The result of `inlist()` or `inrange()` is either 1 when the value specified is in range or in list and 0 otherwise (and thus never missing). So, if you use a variable to store results, let it be a byte variable for efficiency in storage.

In more detail: so long as none of the arguments `z, a, b` is missing, `inrange(z, a, b)` is true whenever `z ≥ a` and `z ≤ b`. Thus `inrange(60, 50, 70)` is true (numerically 1) because `60 ≥ 50` and `60 ≤ 70`. However, `inrange(60, 70, 50)` is false (0) because `60` is not ≥70 and `60` is not ≤50. Thus the order of `a` and `b` is crucial. There are situations when you are not sure in advance about the ordering of arguments, but you can always use devices such as `inrange(z, min(a, b), max(a, b))` (which tests whether one value is between two others).

The definition of `inrange()` is more complicated when any argument is numeric missing. See `functions` for the precise definitions. The most important example is
that \texttt{inrange}(z, a, .) is interpreted as $z \geq a$ and $z < .$ ($z$ greater than or equal to $a$, but not missing). This may look like a bug, but it is really a feature. Even experienced users sometimes forget that in Stata numeric missing is regarded as arbitrarily large. Hence, $z \geq 42$ will be true for all the missing values of $z$, as well as for all values that are greater than or equal to 42. The longstanding workaround when this is not what you want with regard to missing values is to add the extra condition that $z$ is not missing, as in $z \geq 42 \& z < .$, but \texttt{inrange}(z, 42, .) is another way to do this.

The definitions that come into play when any argument is missing imply that \texttt{inrange()} is not a good tool to use when you want to test for numeric missings (including any comparisons with extended missing values). For that it is better to use \texttt{missing()}, \texttt{inlist()}, or combined statements using simple inequalities.

\texttt{inlist()} and \texttt{inrange()} can often be used with the in-built quantities \_n and \_N specifying, respectively, the current observation number and the current number of observations. Sometimes users wish to specify that a command should apply to an irregular set of observation numbers, and \texttt{if inlist(_n,17,42,99,217)} exemplifies how that could be done with a small set (the limit is 255 numbers and is unlikely to bite in sensible practice). A pitfall here is clearly that any sorting of the dataset will often imply that the observations concerned end up in different positions. Thus saving the results of this computation in a byte variable will often be a good idea. This approach is not better general practice than using criteria such as those based on variable values, but there may be occasions when you will want this feature.

Other examples of the same kind arise with longitudinal or panel data. Recently I wanted to identify the first and last values of a response in each panel, and

\begin{verbatim}
   . by panelvar (timevar): gen y_ends = y if inlist(_n, 1, _N)
\end{verbatim}

offers a way to do that. Conversely, \texttt{!inlist(_n, 1, _N)} identifies all the others. Whether you prefer that if condition to the more traditional \texttt{if \_n == 1 \| \_n == \_N} is admittedly a matter of taste. Using \texttt{in} is not an option here because \texttt{in} may not be combined with \texttt{by:}.

The examples so far are all for numeric arguments. The arguments of either function can be all numeric or all string. Thus given one character, \texttt{c}, \texttt{inrange("c", "a", "z")} tests whether \texttt{c} is one of the 26 lowercase letters of the English alphabet; correspondingly, \texttt{inrange("c", "A", "Z")} tests whether \texttt{c} is one of the 26 uppercase letters of the same alphabet. More generally, \texttt{inrange("string", "a", "z")} tests whether \texttt{string} begins with a lowercase letter, and correspondingly for the arguments \texttt{"A", "Z"} and uppercase letters. Because lowercase and uppercase letters are typically not adjacent in your computer’s character sets, be careful when working with both. If you were indifferent about the distinction between uppercase and lowercase, you could work with \texttt{lower("string")} or \texttt{upper("string")}.
These examples lead directly to a way of filtering a string variable to select characters that you want or ignore characters that you don’t. Suppose that we wanted to select only the alphabetic characters in a string variable. Check the variable type to see its maximum length (18, or whatever), `generate` a new empty-string variable, and then loop over the characters, adding them to the end of the new variable only if they are as desired.

```stata
generate newvar = ""
quietly forvalues i = 1/18 {
    replace newvar = newvar + substr(oldvar,'i',1) ///
        if inrange(lower(substr(oldvar,'i',1)),"a","z")
}
```

Commands like this tend to become rather long, but they are not in principle complicated. The attraction of a low-level approach is that you can design exactly the filter you wish according to the precise problem it is intended to solve.

A further simple but general moral evident from various examples here is that the power of Stata functions often arises from how they can be combined.